
Why study diverging programs?

In languages without input/output (IMP, purely functional):

• No practical interest.
• Useful for theory: computability, program equivalences.

In real-world computing:

• Many programs are supposed to never stop! (OS kernels,
Web servers, control-command codes, . . . )

• Reactive divergence: finite computation between successive
I/O operations.

How to formalize divergence?

Negatively:

• Diverging programs are those programs that do not
terminate, neither normally nor by going wrong.

Positively, or even constructively: (parts 2 and 3)

• Coinductive characterizations of divergence.
(Termination is fundamentally inductive..)

At the same time we formalize termination: (parts 1 and 3)

• One of the main goals of denotational semantics.
• Classically (domain theory) or constructively

(partiality monad)

From a bounded interpreter to a
denotational semantics



A reference interpreter

As seen in the first lecture: it is impossible to define the
semantics of an IMP command as a function

store “before”→ store “after”

since this function would be partial (non-termination).

However, we can define an approximation of this function by
bounding a priori the recursion depth, using a fuel parameter of
type nat.

A bounded reference interpreter

Fixpoint cinterp (fuel: nat) (c: com) (s: store)

: option store :=

...

A Some s’ result mean that c is guaranteed to terminate on s’.

A None result is unconclusive: either c diverges, either we need
more fuel to finish the execution of c.

These results form a monad (≈ error monad):

Definition ret {A: Type} (v: A) := Some v.

Definition bind {A B: Type} (x: option A) (f: A -> option B) :=

match x with None => None | Some v => f v end.

A bounded reference interpreter

Fixpoint cinterp (fuel: nat) (c: com) (s: store) : option store :=

match fuel with

| O => None

| S n =>

match c with

| SKIP => Some s

| ASSIGN x a => Some (update x (aeval a s) s)

| SEQ c1 c2 => bind (cinterp n c1 s) (cinterp n c2)

| IFTHENELSE b c1 c2 =>

cinterp n (if beval b s then c1 else c2) s

| WHILE b c1 =>

if beval b s

then bind (cinterp n c1 s) (cinterp n (WHILE b c1))

else Some s

end

end.

Partial order over results

The order r v r′, read “r′ is more defined than r”

None v r′ Some(v) v Some(v)

A crucial property: the interpreter is monotonically increasing.
(More fuel⇒ more defined result.)

i ≤ j ⇒ cinterp i s c v cinterp j s c

Towards a denotational semantics

What happens if “fuel goes to infinity” ?

For a command c that terminates when started in store s:

nNone

Some(s′)

cinterp n c s

For a command c that diverges when started in store s:

nNone

Some(s′)

cinterp n c s

Limits

Every increasing sequence f : nat→ option A has a limit lim f ,
equal to its supremum, and characterized by

∃i, ∀j, i ≤ j⇒ f j = lim f

This claim is not constructive: the Coq development (file
Divergence.v) uses

• the excluded middle axiom to show that the limit exists
(either ∀i, f i = None or ∃i, f i 6= None);

• an axiom of description to define the limit lim f as a function
of the sequence f .

A denotational semantics for IMP

Define [[c]], the denotation of command c, as the limit of c’s
executions by the reference interpreter:

[[c]] s def
= lim(fun n⇒ cinterp n c s)

This definition satisfies the expected equations:

[[skip]] s = Some(s)

[[x := a]] s = Some(s{x← [[a]] s})
[[c1; c2]] s = bind ([[c1]] s) [[c2]]

[[if b then c1 else c2]] s =

[[c1]] s if [[b]] s = true

[[c2]] s if [[b]] s = false

A denotational semantics for IMP: the case of loops

[[while b do c]] s =

bind ([[c]] s) [[while b do c]] if [[b]] s = true

Some(s) if [[b]] s = false

Furthermore, [[while b do c]] is the smallest function
F : store→ option store solution of the equation

F s =

bind ([[c]] s) F if [[b]] s = true

Some(s) if [[b]] s = false

Example: [[while true do c]] s = None because the function
fun s => None is a solution of the equation.

Equivalence between natural and denotational semantics

c/s ⇓ s′ ⇒ [[c]] s = Some(s′)

[[c]] s = Some(s′) ⇒ cinterp n c s = Some(s′) ⇒ c/s ⇓ s′

(Proofs in file Divergence.v.)

Advanced topic: domain theory

A domain is a set A equipped with a partial order v

x v x (reflexive)
x v y ∧ y v z⇒ x v z (transitive)
x v y ∧ y v x⇒ x = y (antisymmetric)

that is ω-complete: every increasing sequence has a supremum.

u0 v u1 v · · · v un v · · · ⇒ sup u ∈ A

(In the literature, this is often called an ω-cpo or just a cpo.)

Examples of domains

Flat domain Pointed domain
(≈ value of a base type) (≈ computation of a base type)

0 1 2 3 4 5 . . .

⊥

0 1 2 3 4 5 . . .

Lazy pairs (≈ the OCaml type bool lazy * bool lazy)

(⊥,⊥)

(⊥, 0) (⊥, 1) (0,⊥) (1,⊥)

(0, 0) (0, 1) (1, 0) (1, 1)

Examples of domains

Stream of Booleans:

⊥

0 :: ⊥ 1 :: ⊥

0 :: 0 :: ⊥ 0 :: 1 :: ⊥ 1 :: 0 :: ⊥ 1 :: 1 :: ⊥

Combinations of domains:

• “Pointing” (adding a minimal element): D⊥ = D ] {⊥}
• Product D1 × D2, sum D1 + D2.
• Continuous functions [D1 → D2].

15



Continuous functions

A function f : D1 → D2 is Scott-continuous if it preserves the
supremum of increasing sequences:

sup f (ui) = f (sup ui)

All continuous functions are increasing. The converse is false.

Example: the function finite stream 7→ 0, infinite stream 7→ 1 is
increasing yet discontinuous. Besides, it is not computable.

Theorem (Scott’s fixed point theorem)

If D is a pointed domain, any continuous function F : D→ D has a
least fixed point µF = supn Fn(⊥).

Recursive types

To interpret recursive data types as domains, we need to solve
(up to isomorphism) equations between domains, such as:

• Integer lists: Dlist ' {nil}+ Nat× Dlist

• Pure lambda-terms: D∞ ' [D∞ → D∞]

This can be done if we use algebraic domains, also called Scott
domains, where every element is the limit of a sequence of
compact elements (≈ finitely described elements).

(See Plotkin’s lecture notes in references.)

Coinductive predicates and natural
semantics for divergence



Predicates defined by axioms and inference rules

P(skip, s)
c/s→ c′/s′ P(c′, s′)

P(c, s)

Up to now, we have interpreted such definitions of predicates in
an inductive manner:

• as the least fixed point of an operator;
• in terms of finite derivations.

Another interpretation exists, the coinductive interpretation:

• as the greatest fixed point of an operator;
• in terms of infinite or finite derivations.

Operator associated with a definition

P(skip, s)
c/s→ c′/s′ P(c′, s′)

P(c, s)

To this axiom and this inference rule, we associate the operator

F(X) = {(skip, s)} ∪ {(c, s) | c/s→ c′/s′ ∧ (c′, s′) ∈ X}

Intuitively: F(X) is the set of all facts that we can deduce by
assuming the facts X and by applying one axiom or one inference
rule.

The F operator is increasing, therefore it has a least fixed point
and a greatest fixed point.

Smallest fixed point

F(X) = {(skip, s)} ∪ {(c, s) | c/s→ c′/s′ ∧ (c′, s′) ∈ X}

The smallest fixed point is

µF def
=

⋂
{X | F(X) ⊆ X}

It is the limit of the increasing sequence ∅, F(∅), . . . , Fn(∅), . . .

In the example, Fn(∅) is the set of (c, s) that reduce to skip in at
most n reductions. Hence, µF is the set of (c, s) that terminate
(c/s ∗→ skip/s′).

Finite derivations

A derivation = a tree with axioms at the leaves and inference
rules at the nodes.

The inductive interpretation µF corresponds to facts that are
conclusion of a derivation tree where all branches are finite.

(If all rules have finitely many premises, these derivations are the
finite trees.)

Finite derivations

An example of a finite derivation:

c1/s1 → c2/s2

c2/s2 → c3/s3

cn/sn → skip/sn+1

...

P(c3, s3)

P(c2, s2)

P(c1, s1)

P(c, s) can be derived by a tree of height n if and only if c/s
reduces to skip in n step.

Greatest fixed point

F(X) = {(skip, s)} ∪ {(c, s) | c/s→ c′/s′ ∧ (c′, s′) ∈ X}

The greatest fixed point is

νF def
=

⋃
{X | X ⊆ F(X)}

It’s the limit of the decreasing sequence U, F(U), . . . , Fn(U), . . .
where U is the universe of all pairs (c, s).

In the example, νF comprises

• all the (c, s) that terminate: c/s ∗→ skip/s′

• all the (c, s) that diverge: c/s ∗→ cn/sn → · · ·

Infinite derivations

The coinductive interpretation νF corresponds to the facts that
are conclusion of a finite or infinite derivation tree.

An example of infinite derivation:

c1/s1 → c2/s2

c2/s2 → c3/s3

cn/sn → cn+1/sn+1
...

...

P(c3, s3)

P(c2, s2)

P(c1, s1)

Divergence, co-inductively

c/s→ c′/s′ div(c′, s′)

div(c, s)

Inductive interpretation: always false! (there are no axioms. . . )

Coinductive interpretation (double horizontal line):
characterizes the existence of an infinite sequence of reductions.

In Coq:

CoInductive div: com * state -> Prop :=

| div_intro: forall c s c’ s’,

red (c, s) (c’, s’) -> div (c’, s’) ->

div (c, s).

Coinduction principle

By definition of the greatest fixed point νF = ∪{X | X ⊆ F(X)},
any X such that X ⊆ F(X) is contained in νF.

Hence: if the predicate X : com ∗ store→ Prop satisfies

∀c,∀s, X (c, s)⇒ ∃c′, ∃s′, c/s→ c′/s′ ∧ X (c′, s′)

then X (c, s) implies div (c, s).

Graphically:

X:
•

•
• • •

Coinduction principle

From this coinduction principe, we can derive another, very
useful principle, using +→ (one or several reductions) instead of
→ (one reduction):

If ∀c, ∀s, X c s⇒ ∃c′,∃s′, c/s +→ c′/s′ ∧ X c′ s′,
then X c s implies div c s.

X:
•

•
•

• •

• •

Back to natural semantics

In the first lecture, we introduced natural semantics as a way to
structure the reductions to skip.

For example, if command c; c′ terminates, its reduction sequence
must have the following shape:

(c; c′)/s→ (c1; c′)/s1 → · · · → (skip; c′)/s′

→ c′/s′ → · · · → skip/s′′

This structure is reflected by the natural semantics rule for
sequences:

c/s ⇓ s′ c′/s′ ⇓ s′′

c; c′/s ⇓ s′′

28



A structure for infinite sequences of reductions

Likewise, if command c; c′ diverges, its infinite sequence of
reductions must have one of the following two shapes:

(c; c′)/s→ · · · → (cn; c′)/sn → · · · (1)

(c; c′)/s ∗→ (skip; c′)/s′ → c′/s′ → · · · c′n/s′n → · · · (2)

In case (1): c diverges, c′ does not get to run.
In case (2): c terminates, then c′ diverges.

Let’s try to reflect this structure as rules for a predicate c/s ⇑ ,
“command c diverges started in store s”.

29



Natural semantics for divergence

c1/s ⇑

c1; c2/s ⇑

c1/s ⇓ s′ c2/s′ ⇑

c1; c2/s ⇑

[[b]] s = true c1/s ⇑

(if b then c1 else c2)/s ⇑

[[b]] s = false c2/s ⇑

(if b then c1 else c2)/s ⇑

[[b]] s = true c/s ⇑

(while b do c)/s ⇑

[[b]] s = true c/s ⇓ s′ (while b do c)/s′ ⇑

(while b do c)/s ⇑

An example of diverging execution

The loop c def
= while true do x := x+ 1 diverges.

x := x + 1/s0 ⇓ s1

x := x + 1/s1 ⇓ s2

x := x + 1/s2 ⇓ s3
...

c/s2 ⇑

c/s1 ⇑

c/s0 ⇑

(Where si = s0[x← s0(x) + i].)

Equivalence with reduction semantics

Theorem

If c/s ⇑ , then c/s reduces infinitely.

Proof (constructive).

We show c/s ⇑ ⇒ ∃c′,∃s′, c/s +→ c′/s′ ∧ c′/s′ ⇑ .
We conclude by the second coinduction principle applied to the set
X = {(c, s) | c/s ⇑ }.

Theorem

If c/s reduces infinitely, then c/s ⇑ .

Proof (classical).

By coinduction and case analysis on the shape of reduction sequences.
We need excluded middle: either c/s reduces finitely to skip, or c/s
reduces infinitely.

Application to compiler verification

In lecture #2, we used natural semantics to show the correctness
of the compiled code for a terminating IMP command:

Lemma compile_com_correct_terminating:

forall s c s’, cexec s c s’ ->

forall C pc σ, code_at C pc (compile_com c) ->

transitions C

(pc, σ, s)

(pc + codelen (compile_com c), σ, s’).

An induction on the derivation of cexec s c s’ led to a rather
simple proof.

Paradise lost: this simple proof did not extend to diverging
commands.

Application to compiler verification

Paradise regained: natural coinductive semantics leads to a
rather simple proof of compiler correctness for diverging
commands.

Consider the set of machine configurations corresponding to
diverging commands:

X def
= {(pc, σ, s) | ∃c, c/s ⇑ ∧ code at C pc (compile com c)}

We show that this set is “productive”:

∀(pc, σ, s) ∈ X, ∃(pc′, σ′, s′) ∈ X, (pc, σ, s) +→ (pc′, σ′, s′)

We conclude that, when started in a configuration that belongs to
X, the machine performs infinitely many transitions.

Partiality monad and coinductive
reference interpreter



Partial computations in type theory

(V. Capretta, General recursion via coinductive types, LMCS(1), 2005)

CoInductive delay (A: Type) : Type :=

| now: A -> delay A

| later: delay A -> delay A.

delay A represents computations that produce a value of type A
if they terminate.

The later constructor materializes one step of computation.

The delay type being coinductive, we can have infinitely many
steps of computation, that is, a non-terminating computation.

CoFixpoint bottom (A: Type) : delay A := later (bottom A).

Partial computations

CoInductive delay (A: Type) : Type :=

| now: A -> delay A

| later: delay A -> delay A.

Terminating computations are characterized inductively;
diverging computations, coinductively.

Inductive terminates (A: Type) : delay A -> A -> Prop :=

| terminates_now:

forall v, terminates (now v) v

| terminates_later:

forall a v, terminates a v -> terminates (later a) v.

CoInductive diverges (A: Type) : delay A -> Prop :=

| diverges_later:

forall a, diverges a -> diverges (later a).

General recursion

We can define general recursive functions with delay result type,
provided all recursive calls are guarded by later.

8 Fixpoint remainder (a b: nat) : nat :=

if a <? b then a else remainder (a - b) b.

8 CoFixpoint remainder (a b: nat) : delay nat :=

if a <? b then now a else remainder (a - b) b.

4 CoFixpoint remainder (a b: nat) : delay nat :=

if a <? b then now a else later (remainder (a - b) b).

Recursion and corecursion in Coq

Recursive function definition (Fixpoint):

• The argument has an inductive type.
• Guard condition: f x can call f y recursively provided the

argument y is a strict sub-term of the argument x.

Corecursive function definition (CoFixpoint):

• The result has a coinductive type.
• Productivity condition: f x can call f y recursively provided

the result f y is a strict sub-term of the result f x.
(f x is f y wrapped inside one or several constructors.)

General recursion

CoFixpoint remainder (a b: nat) : delay nat :=

if a <? b then now a else later (remainder (a - b) b).

We can reason about termination or divergence of the function
after its definition.

Theorem remainder_Euclid:

forall a b, b > 0 ->

exists q r, terminates (remainder a b) r ∧ r < b ∧ a = b*q+r.

Theorem remainder_divergence:

forall a, diverges (remainder a 0).

Observational equivalence

A constructive definition of equitermination:

CoInductive equi {A: Type} : delay A -> delay A -> Prop :=

| equi_terminates: forall x y v,

terminates x v -> terminates y v -> equi x y

| equi_later: forall x y,

equi x y -> equi (later x) (later y).

In classical logic, this is equivalent to

(∃v, terminates x v∧terminates y v)∨ (diverges x∧diverges y)

but it is stronger in constructive logic. (No need to “guess in
advance” whether the two computations terminate or diverge.)

The partiality monad

The delay type is a monad, with constructor now as the ret

operation, and the bind operation defined as the sequencing of
two computations.

CoFixpoint bind (A B: Type)

(a: delay A) (f: A -> delay B) : delay B :=

match a with

| now v => later (f v)

| later a’ => later (bind a’ f)

end.

We have the expected properties for a sequencing, e.g. bind a f

diverges i� a diverges or a terminates with v and f v diverges.

The partiality monad

The three monadic laws hold up to observational equivalence
(equi, written ≈ from now on):

bind (now v) f ≈ f v

bind a now ≈ a

bind (bind a f ) g ≈ bind a (fun x⇒ bind (f x) g)

Furthermore,

bind a f ≈ bind a′ f ′ if a ≈ a′ and ∀x, f x ≈ f ′ x

An interpreter in the partiality monad

Using the partiality monad, let’s try to write a general recursive
interpreter for IMP.

CoFixpoint cinterp (c: com) (s: store) : delay store :=

match c with

| SKIP => now s

| ASSIGN x a => now (update x (aeval a s) s)

| SEQ c1 c2 => bind (cinterp c1 s) (cinterp c2)

| IFTHENELSE b c1 c2 =>

later (cinterp (if beval b s then c1 else c2) s)

| WHILE b c =>

if beval b s then bind (cinterp c s) (cinterp (WHILE b c))

else ret s

end.

Problem: this definition is not productive!

Going through the free monad

(An application of N. A. Danielsson’s technique, Beating the Productivity Checker

Using Embedded Languages, 2010)

We can work around the problem by presenting the monad as a
coinductive type whose constructors are the monad operations:
ret, bind, and later.

CoInductive mon: Type -> Type :=

| Ret: forall {A: Type}, A -> mon A

| Later: forall {A: Type}, mon A -> mon A

| Bind: forall {A B: Type}, mon A -> (A -> mon B) -> mon B

In other words: the free monad (plus later).

In other words: an abstract syntax for Moggi’s monadic
metalanguage (plus later).

An interpreter in the free monad

CoFixpoint cinterp (c: com) (s: store) : mon store :=

match c with

| SKIP => Ret s

| ASSIGN x a => Ret (update x (aeval a s) s)

| SEQ c1 c2 => Bind (cinterp c1 s) (cinterp c2)

| IFTHENELSE b c1 c2 =>

Later (cinterp (if beval b s then c1 else c2) s)

| WHILE b c =>

if beval b s then Bind (cinterp c s) (cinterp (WHILE b c))

else Ret s

end.

This definition is productive!

Interpreting the free monad

A term of type mon A describes a computation of type delay A.

CoFixpoint run {A: Type} (m: mon A) : delay A :=

match m with

| Ret v => now v

| Later m => later (run m)

| Bind (Ret v) f => later (run (f v))

| Bind (Later m) f => later (run (Bind m f))

| Bind (Bind m f) g =>

later (run (Bind m (fun x => Bind (f x) g)))

end.

Note the use “on the fly” of the first and third monadic laws.

Productivity is surprising

com * store

mon store

delay store

cinterp 4

run 4

8 cinterp

The productivity condition is a syntactic approximation.
It is not compositional.

The run function as a denotational semantics

The run function can be viewed as a denotational semantics for
the monadic metalanguage:

run : syntax (type mon A)→ meaning (type delay A).

Equivalences satisfied by run:

run (Later m) ≈ later(run m) (Later denotation)

run (Bind m f ) ≈ bind (run m) (fun x⇒ run (f x))
(Bind denotation)

run (Bind (Ret v) f ) ≈ run (f v) (first monadic law)

run (Bind m Ret) ≈ run m (second monadic law)

run (Bind (Bind m f ) g) ≈ run (Bind m (fun x⇒ Bind (f x) g))
(third monadic law)

The coinductive interpreter as a denotational semantics

Define the denotation of a command c as

[[c]] s def
= run (cinterp c s) (with type delay store)

This definition satisfies the expected equations:

[[skip]] s ≈ now(s)

[[x := a]] s ≈ now(s{x← [[a]] s})
[[c1; c2]] s ≈ bind ([[c1]] s) [[c2]]

[[if b then c1 else c2]] s ≈

{
[[c1]] s if [[b]] s = true

[[c2]] s if [[b]] s = false

[[while b do c]] s ≈

{
bind ([[c]] s) [[while b do c]] if [[b]] s = true

now(s) if [[b]] s = false

Summary



Summary

Coinduction is a fundamental tool to reason about divergence,
from the trivial (infinite sequences of reductions) to the subtle
(natural semantics for divergence, partiality monad).

Denotational semantics require an appropriate mathematical
structure. Classically, it’s Scott domains; constructively, it could
be the quotient type delay A/ ≈.

For the time being, the approaches outlined in this lecture do not
scale to “big languages” as well as transition semantics.

